
ABSTRACT
The COMPARE procedure is very useful for validating SAS® data sets in clinical studies. One particularly useful
feature is its SYSINFO system macro variable by which numerical codes are issued following the comparison of data
sets. These codes represent up to 16 messages that describe the outcome of the comparisons. Thus, interpreting
SYSINFO codes can provide a very concise PROC COMPARE report, especially when many pairs of data sets are
involved. In this paper, we demonstrate a novel approach for decoding SYSINFO values using the SAS function
bAND (bitwise logical AND). With this function, one can quickly determine which of the 16 PROC COMPARE output
messages a SYSINFO code contains.

INTRODUCTION
An essential part of establishing the integrity of clinical trial data is the validation of the data sets used to generate the
listings, tables, and graphs in Clinical Study Reports (CSR). To achieve that goal, the SAS® Compare procedure
becomes a very convenient tool.
However, the standard PROC COMPARE reports are often overwhelming and cumbersome. Results of comparison
can run into many pages and can prove challenging when one is handling dozens of folders containing hundreds of
data sets.
Rather fortunately, PROC COMPARE also produces a compact reporting system by issuing return codes based on
the outcome of the comparisons. These return codes are stored in the automatic system macro variable called
SYSINFO. In this paper, we present a novel approach of interpreting SYSINFO codes using the SAS® bitwise
function called "bAND" or "bitwise AND". This is possible because SYSINFO codes are binary in nature.

SYSINFO CODES
The SYSINFO return codes represent binary numbers.
PROC COMPARE looks for 16 possible differences between data sets (see Table 1).
These 16 differences can be viewed as 16 bit positions in a binary number of 16 digits:

where 0 at a particular position means the difference is absent and 1 means the difference has been detected.
For example, 1 at bit position 12 means Difference#12 was found in the comparison:

where Difference#12 is: "Comparison data set has variable not in Base".
The binary code for 0000100000000000 is 2048 (decimal form).
Thus when SYSINFO code is 2048, it means "Comparison data set has variable not in Base".
That is the basis for interpreting SYSINFO codes.

When many of the 16 differences are found in a single comparison, the binary codes are added.
Thus, SYSINFO=14369 decodes to: "0011100000100001" in binary,
which can be interpreted as:
"Data set labels differ"
"Variable has different label"
"Comparison data set has variable not in Base"
"A value comparison was unequal"
"Conflicting variable types"

(1st bit position =1)
(6th bit position =1)
th
(12 bit position =1)
(13th bit position =1)
th
(14 bit position =1)

So a single decimal number provided by SYSINFO can hold several messages, making reporting very compact. But there is still
the drudgery of obtaining the messages hidden in those codes.
Table 1: The Meaning of SYSINFO Codes

2

SAS Global Forum 2012

Coders' Corner

INTERPRETING SYSINFO CODES
There are many easy ways to interpret SYSINFO return codes.
(a) The easiest and simplest but inefficient way is to use just IF statements:
if sysinfo=0 then message="There is no difference between Base data set and Comparison";
if sysinfo=64 then message="Base data set has observation not in Comparison";
if sysinfo=32768 then message ="Fatal error: comparison not done";
(b) Using IF statements with Bit Masks:
This approach is quite efficient although it requires sixteen IF statements.
SYSINFO codes, being binary numbers, can be subject to "bit testing", the process of detecting 1's in particular positions in a
binary number. A "bit mask" is a SAS® binary literal used for bit testing. The binary literal, '………….1'b , can be used as a bit
mask in a bit test. A bit test is the comparison of the binary form of a number against a bit mask. The periods in a bit mask are
neutral place holders and yield a TRUE result regardless of the value of the corresponding bit.
In other words, if a binary number is: 000100110100, and the bit mask is "…………1..", that bit-mask would detect just the "1" in
rd
the 3 position from the right, ignoring all other 1's and 0's elsewhere. Bit masks therefore detect the presence of "1" in a particular
position in a binary number.
To interpret a SYSINFO code, a bit mask is used as follows:
if sysinfo='1'b then message= "Data set labels differ";
if sysinfo='1.'b then message= "Data set types differ";
if sysinfo='1..'b then message= "Variable has different informat";
if sysinfo='1…'b then message= "Variable has different format";
if sysinfo='1….'b then message= "Variable has different length";
if sysinfo='1…..'b then message= "Variable has different label";
and so on.

THE BITWISE-AND (BAND) FUNCTION
SAS® provides six bitwise logical functions: bAND, bLSHIFT, bNOT, bOR, bRSHIFT, and bXOR.
The bAND function returns the bitwise logical AND of two arguments.
The syntax is: bAND (argument-1, argument-2), where the arguments can be numbers, variables, or expressions.

SYSINFO INTERPRETATION WITH BAND
The nice thing about bAND which is exploited for SYSINFO decoding is this:
if a binary code x contains another binary code y, then bAND (x, y) = y, else bAND (x, y) = 0.
For example, earlier we showed that the SYSINFO code 14369 contains 4 messages coded by 1, 32, 2048, and 4096 respectively.
Thus,
bAND (14369, 1) = 1
bAND (14369, 32) = 32
bAND (14369, 2048) = 2048
bAND (14369, 4096) = 4096
By contrast,
bAND (14369, 2) = 0
bAND (14369, 512) =0
because SYSINFO 14369 does not code for "Data set types differ" (code 2),
nor does it code for "Comparison data set has BY group not in Base" (code 512).
Thus the bAND function is pretty effective for unraveling the messages embedded in SYSINFO return codes.

DATA SET VALIDATION PROGRAM USING BAND
The use of the bAND function with SYSINFO return codes enables the efficient validation of a large set of SAS® tables. In fact, the
validation program we have developed, "ValidateDatasets.sas", compares folders containing large numbers of data sets, producing
a single table of report for each folder, by means of SYSINFO codes and their meanings.
The program is organized as follows:
[SECTION-A]

INPUT DATA PREPARATION:
Step 01: Create macro variables for file pathnames
Step 02: Pull data set names from dictionary.tables into macro variables

[SECTION-B]

PROC COMPARE PROCESSING
Step 03: Prepare a table (syscodes) to receive PROC COMPARE sysinfo codes
Step 04: Run PROC COMPARE to match OLD data sets with NEW data sets

[SECTION-C]

SYSINFO CODE PROCESSING
Step 05: Capture PROC COMPARE SYSINFO codes into macro variables
Step 06: Decode each SYSINFO code into a PROC COMPARE message
Step 07: Accumulate in SYSINFO table (syscodes) data set names,
SYSINFO codes, and decoded messages
Step 08: Call Macro

[SECTION-D]

OUTPUT
Step 09: Print SYSINFO report from table (syscodes)

*------------------------------------------------------------------------------------------------------------[SECTION-A]
I N P U T
D A T A
P R E P A R A T I O N
-------------------------------------------------------------------------------------------------------------;
*----------------Step 01: Create macro variables for file pathnames------------------------------------------;
%let datapath=C:\Documents and Settings\hinsonj\Desktop\Publication;
%let study=123;
libname OLD&study "C:\Documents and Settings\hinsonj\Desktop\Publication\OLD123";
libname NEW&study "C:\Documents and Settings\hinsonj\Desktop\Publication\NEW123";
*---------- Step 02: Pull data set names from dictionary.tables into macro variables ------------------------;
proc sql noprint;
select memname into :OLDset1 - :OLDset&sysmaxlong
from dictionary.tables
where indexw("OLD&study" , libname);
select memname into :NEWset1 - :NEWset&sysmaxlong
from dictionary.tables
where indexw("NEW&study" , libname);
quit;
%let nsize=&sqlobs;
*------------------------------------------------------------------------------------------------------------[SECTION-B]
P R O C
C O M P A R E
P R O C E S S I N G
-------------------------------------------------------------------------------------------------------------;
%macro sys0proc0compare;
%*-------------Step 03:
Prepare a table to receive Proc Compare sysinfo codes -----------------------------;
proc sql;
create table syscodes
(DATASET char(14), SYSINFO num, MEANING_OF_SYSINFO_CODES char(400));
quit;
%*-------------Step 04:
Run Proc Compare to match OLD data sets with NEW data sets-------------------------;
%do N=1 %to &nsize;
proc compare
base=OLD&study..&&OLDset&N.
compare=NEW&study..&&NEWset&N.
;
run;
%*------------------------------------------------------------------------------------------------------------

[SECTION-C]
S Y S I N F O
C O D E S
P R O C E S S I N G
--------------------------------------------------------------------------------------------------------------;
%*-------------Step 05:
Capture Proc Compare SYSINFO codes into macro variables ----------------------------;
%let sicode=&sysinfo;
%let dsname=&&NEWset&N;
%*-------------Step 06:
Decode each SYSINFO code into a Proc Compare message -------------------------------;
data _null_;
length message $ 600 decoded $ 600 text $60;
array msg {17} $ 60 _temporary_ (
" ",
"Data set labels differ",
"Data set types differ",
"Variable has different informat",
"Variable has different format",
"Variable has different length",
"Variable has different label",
"OLD data set has observation not in NEW",
"NEW data set has observation not in OLD",
"OLD data set has BY group not in NEW",
"NEW data set has BY group not in OLD",
"OLD data set has variable not in NEW",
"NEW data set has variable not in OLD",
"A value comparison was unequal",
"Conflicting variable types",
"BY variables do not match",
"Fatal error: comparison not done"
);
testcode=&sicode;
testcode=&sicode;
if testcode=0 then decoded="NO DIFFERENCE BETWEEN OLD & NEW"; *<---(&SYSINFO=0
WHEN NO DIFFERENCE DETECTED);
else do;
decoded=" "; *<----------(VARIABLE 'DECODED' WILL STORE &SYSINFO MESSAGES);
do k=1 to 16; *<----------(BECAUSE THERE ARE 16 POSSIBLE &SYSINFO MESSAGES);
binval=2**(k-1); *<--(CONVERT 0,1,2,3...16 TO BINARY 1,2,4,8,16,32,64..);
match=band(binval, testcode); *<--(DO BITWISE TESTING WITH BAND FUNCTION);
key=sign(match)*k; *<---(REVERT BINARIES TO REGULAR NUMBERS FOR ARRAYINDEXING);
text=msg(key+1); *<---(GET DECODED MESSAGE TEXT FROM ARRAY VALUE);
decoded=catx(", ",decoded,text); *<---(CONCATENATE IF MORE THAN ONE &SYSINFO
MESSAGE);
end;
end;
call symputx("message",decoded);
run;

%*----Step 07:

Accumulate in SYSINFO table(syscodes) data set names, SYSINFO codes, and decoded messages --;
proc sql;
insert into syscodes values("&dsname.",&sicode.,"&message.");
quit;

%end;
%mend sys0proc0compare;
*----Step 08:
Call Macro ---------------------------------------------------------------------------------;
%sys0proc0compare;
*------------------------------------------------------------------------------------------------------------[SECTION-D]
O U T P U T
-------------------------------------------------------------------------------------------------------------;
*----Step 09:
Print SYSINFO report from table(syscodes)---------------------------------------------------;
ODS RTF FILE="C:\Documents and Settings\hinsonj\Desktop\Publication\syscodes.rtf";
ODS NOPROCTITLE;
proc print data=syscodes;
title "Comparisons Between OLD and NEW data sets: A Sysinfo Report";
run;
ODS RTF CLOSE;
*_ _ _ _ _ _ _ _ _ _ _ _ END OF PROGRAM ValidateDatasets.sas _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ ;

Table 2: Folders of Data Sets to Validate

Table 3: Proc Compare SYSINFO-Based Report

CONCLUSION
We have demonstrated in this paper yet another technique for validating data sets underlying clinical trial data.
The use of the SAS® bitwise AND function enables an efficient interpretation of the SYSINFO return codes generated
by PROC COMPARE, and allows the generation of compact comparison reports, as part of clinical data validation.
This approach also enables the automatic insertion of reports into Excel workbook by outputting SYSINFO reports
into Excel worksheets where a single worksheet represents an entire folder of SAS® data sets (not shown in this
paper).
The SYSINFO approach further permits the programming of decision making when certain differences are detected
among data sets. For instance, through programming logic, one can choose to ignore "cosmetic" differences between
data sets. SYSINFO codes less than 64 usually relate to formatting and not considered critical.
Similarly, a comparison that generates SYSINFO=0 for data sets might not even need to generate reports, as the
return code would indicate "no differences detected". A code of 32768 ("Fatal error") could be made to trigger the
suspension of the running of the program, saving valuable time.
It is therefore apparent that SYSINFO coding and interpretation could be a powerful tool in the potentially arduous
task of validating clinical data. The use of the bAND function makes the process even more efficient.

